

**melogit postestimation** — Postestimation tools for melogit

[Postestimation commands](#)
[predict](#)
[margins](#)
[Remarks and examples](#)
[Methods and formulas](#)
[References](#)
[Also see](#)

## Postestimation commands

The following postestimation commands are of special interest after `melogit`:

Command	Description
<code>estat group</code>	summarize the composition of the nested groups
<code>estat icc</code>	estimate intraclass correlations
<code>estat sd</code>	display variance components as standard deviations and correlations

The following standard postestimation commands are also available:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of estimates
<code>estat ic</code>	Akaike's and Schwarz's Bayesian information criteria (AIC and BIC)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estat (svy)</code>	postestimation statistics for survey data
<code>estimates</code>	cataloging estimation results
* <code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of coefficients
* <code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
<code>predict</code>	predictions, residuals, influence statistics, and other diagnostic measures
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of estimates
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

\* `hausman` and `lrtest` are not appropriate with `svy` estimation results.

# predict

## Description for predict

`predict` creates a new variable containing predictions such as mean responses; linear predictions; density and distribution functions; standard errors; and Pearson, deviance, and Anscombe residuals.

## Menu for predict

Statistics > Postestimation

## Syntax for predict

*Syntax for obtaining predictions of the outcome and other statistics*

```
predict [type] newvarsspec [if] [in] [, statistic options]
```

*Syntax for obtaining estimated random effects and their standard errors*

```
predict [type] newvarsspec [if] [in], reffects [re_options]
```

*Syntax for obtaining ML scores*

```
predict [type] newvarsspec [if] [in], scores
```

*newvarsspec* is *stub\** or *newvarlist*.

<i>statistic</i>	Description
------------------	-------------

Main	
<code>mu</code>	mean response; the default
<code>eta</code>	fitted linear predictor
<code>xb</code>	linear predictor for the fixed portion of the model only
<code>stdp</code>	standard error of the fixed-portion linear prediction
<code>density</code>	predicted density function
<code>distribution</code>	predicted distribution function
<code>pearson</code>	Pearson residuals
<code>deviance</code>	deviance residuals
<code>anscombe</code>	Anscombe residuals

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

<i>options</i>	Description
Main	
<code>conditional(<i>ctype</i>)</code>	compute <i>statistic</i> conditional on estimated random effects; default is <code>conditional(ebmeans)</code>
<code>marginal</code>	compute <i>statistic</i> marginally with respect to the random effects
<code>nooffset</code>	make calculation ignoring offset or exposure
Integration	
<code>int_options</code>	integration options
pearson, deviance, anscombe may not be combined with marginal.	
<i>ctype</i>	Description
<code>ebmeans</code>	empirical Bayes means of random effects; the default
<code>ebmodes</code>	empirical Bayes modes of random effects
<code>fixedonly</code>	prediction for the fixed portion of the model only
<i>re_options</i>	Description
Main	
<code>ebmeans</code>	use empirical Bayes means of random effects; the default
<code>ebmodes</code>	use empirical Bayes modes of random effects
<code>reses(<i>stub*</i>   <i>newvarlist</i>)</code>	calculate standard errors of empirical Bayes estimates
Integration	
<code>int_options</code>	integration options
<i>int_options</i>	Description
<code>intpoints(#)</code>	use # quadrature points to compute marginal predictions and empirical Bayes means
<code>iterate(#)</code>	set maximum number of iterations in computing statistics involving empirical Bayes estimators
<code>tolerance(#)</code>	set convergence tolerance for computing statistics involving empirical Bayes estimators

## Options for predict

### Main

`mu`, the default, calculates the predicted mean, that is, the probability of a positive outcome.  
`eta`, `xb`, `stdp`, `density`, `distribution`, `pearson`, `deviance`, `anscombe`, `scores`, `conditional()`, `marginal`, and `nooffset`; see [ME] [meglm postestimation](#).  
`reffects`, `ebmeans`, `ebmodes`, and `reses()`; see [ME] [meglm postestimation](#).

### Integration

`intpoints()`, `iterate()`, and `tolerance()`; see [ME] [meglm postestimation](#).

# margins

## Description for margins

`margins` estimates margins of response for mean responses and linear predictions.

## Menu for margins

Statistics > Postestimation

## Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
<code>mu</code>	mean response; the default
<code>eta</code>	fitted linear predictor
<code>xb</code>	linear predictor for the fixed portion of the model only
<code>stdp</code>	not allowed with <code>margins</code>
<code>density</code>	not allowed with <code>margins</code>
<code>distribution</code>	not allowed with <code>margins</code>
<code>pearson</code>	not allowed with <code>margins</code>
<code>deviance</code>	not allowed with <code>margins</code>
<code>anscombe</code>	not allowed with <code>margins</code>
<code>reflects</code>	not allowed with <code>margins</code>
<code>scores</code>	not allowed with <code>margins</code>

Options `conditional(ebmeans)` and `conditional(ebmodes)` are not allowed with `margins`.

Option `marginal` is assumed where applicable if `conditional(fixedonly)` is not specified.

Statistics not allowed with `margins` are functions of stochastic quantities other than  $e(b)$ .

For the full syntax, see [R] [margins](#).

## Remarks and examples

[stata.com](http://www.stata.com)

Various predictions, statistics, and diagnostic measures are available after fitting a logistic mixed-effects model with `melogit`. For the most part, calculation centers around obtaining estimates of the subject/group-specific random effects. Random effects are not provided as estimates when the model is fit but instead need to be predicted after estimation. Calculation of intraclass correlations, estimating the dependence between latent linear responses for different levels of nesting, may also be of interest.

### ► Example 1: Estimating the intraclass correlation

Following [Rabe-Hesketh and Skrondal \(2012, chap. 10\)](#), we consider a two-level mixed-effects model for onycholysis (separation of toenail plate from nail bed) among those who contract toenail fungus. The data are obtained from [De Backer et al. \(1998\)](#) and were also studied by [Lesaffre and Spiessens \(2001\)](#). The onycholysis outcome is dichotomously coded as 1 (moderate or severe onycholysis) or 0 (none or mild onycholysis). Fixed-effects covariates include treatment (0: itraconazole; 1: terbinafine), the month of measurement, and their interaction.

We fit the two-level model with `melogit`:

```
. use http://www.stata-press.com/data/r15/toenail
(Onycholysis severity)
. melogit outcome treatment month trt_month || patient:, intpoints(30)
(iteration log omitted)
Mixed-effects logistic regression           Number of obs   =       1,908
Group variable:          patient           Number of groups =        294
                                           Obs per group:
                                           min =           1
                                           avg =          6.5
                                           max =           7
Integration method: mvaghermite            Integration pts. =         30
Log likelihood = -625.38557                 Wald chi2(3)    =       150.61
                                           Prob > chi2     =        0.0000
```

outcome	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
treatment	-.1608934	.5802058	-0.28	0.782	-1.298076	.9762891
month	-.3911056	.0443906	-8.81	0.000	-.4781097	-.3041016
trt_month	-.1368286	.0680213	-2.01	0.044	-.2701479	-.0035093
_cons	-1.620355	.4322382	-3.75	0.000	-2.467526	-.7731834
patient						
var(_cons)	16.0841	3.062625			11.07431	23.36021

```
LR test vs. logistic model: chibar2(01) = 565.24      Prob >= chibar2 = 0.0000
```

It is of interest to determine the dependence among responses for the same subject (between-subject heterogeneity). Under the latent-linear-response formulation, this dependence can be obtained with the intraclass correlation. Formally, in a two-level random-effects model, the intraclass correlation corresponds to the correlation of latent responses within the same individual and also to the proportion of variance explained by the individual random effect.

We use `estat icc` to estimate the residual intraclass correlation:

```
. estat icc
```

```
Residual intraclass correlation
```

Level	ICC	Std. Err.	[95% Conf. Interval]	
patient	.8301913	.0268433	.7709672	.8765531

In the presence of fixed-effects covariates, `estat icc` reports the residual intraclass correlation, which is the correlation between latent linear responses conditional on the fixed-effects covariates.

Conditional on treatment and month of treatment, we estimate that latent responses within the same patient have a large correlation of approximately 0.83. Further, 83% of the variance of a latent response is explained by the between-patient variability.

◀

### ► Example 2: Predicting random effects

In [example 3](#) of [\[ME\] melogit](#), we represented the probability of contraceptive use among Bangladeshi women by using the model (stated with slightly different notation here)

$$\text{logit}(\pi_{ij}) = \beta_0 \text{rural}_{ij} + \beta_1 \text{urban}_{ij} + \beta_2 \text{age}_{ij} + \beta_3 \text{child1}_{ij} + \beta_4 \text{child2}_{ij} + \beta_5 \text{child3}_{ij} + a_j \text{rural}_{ij} + b_j \text{urban}_{ij}$$

where  $\pi_{ij}$  is the probability of contraceptive use,  $j = 1, \dots, 60$  districts,  $i = 1, \dots, n_j$  women within each district, and  $a_j$  and  $b_j$  are normally distributed with mean 0 and variance–covariance matrix

$$\Sigma = \text{Var} \begin{bmatrix} a_j \\ b_j \end{bmatrix} = \begin{bmatrix} \sigma_a^2 & \sigma_{ab} \\ \sigma_{ab} & \sigma_b^2 \end{bmatrix}$$

```

. use http://www.stata-press.com/data/r15/bangladesh
(Bangladesh Fertility Survey, 1989)
. generate byte rural = 1 - urban
. melogit c_use rural urban age child*, noconstant || district: rural urban,
> noconstant cov(unstructured)
(iteration log omitted)
Mixed-effects logistic regression          Number of obs    =    1,934
Group variable:          district          Number of groups =     60
                                   Obs per group:
                                   min =           2
                                   avg =          32.2
                                   max =          118

Integration method: mvaghermite           Integration pts. =     7
                                   Wald chi2(6)    =    120.24
                                   Prob > chi2     =     0.0000

Log likelihood = -1199.315
( 1) [c_use]_cons = 0

```

c_use	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
rural	-1.711652	.1605617	-10.66	0.000	-2.026347	-1.396956
urban	-.8958623	.1704954	-5.25	0.000	-1.230027	-.5616974
age	-.026415	.008023	-3.29	0.001	-.0421398	-.0106903
child1	1.13252	.1603285	7.06	0.000	.8182819	1.446758
child2	1.357739	.1770522	7.67	0.000	1.010724	1.704755
child3	1.353827	.18288	7.40	0.000	.9953883	1.712265
_cons	0	(omitted)				
<hr/>						
district						
var(rural)	.3897485	.1292403			.2034823	.7465212
var(urban)	.2442899	.1450625			.0762871	.7822759
<hr/>						
district						
cov(rural, urban)	-.0161411	.1057462	-0.15	0.879	-.2233999	.1911177

```

LR test vs. logistic model: chi2(3) = 58.42          Prob > chi2 = 0.0000
Note: LR test is conservative and provided only for reference.

```

The purpose of using this particular model was to allow for district random effects that were specific to the rural and urban areas of that district and that could be interpreted as such. We can obtain predictions of these random effects and their corresponding standard errors,

```

. predict re_rural re_urban, reffects reses(se_rural se_urban)
(calculating posterior means of random effects)
(using 7 quadrature points)

```

The order in which we specified the variables to be generated corresponds to the order in which the variance components are listed in `melogit` output. If in doubt, a simple `describe` will show how these newly generated variables are labeled just to be sure.

Having generated estimated random effects and standard errors, we can now list them for the first 10 districts:

```
. by district, sort: generate tolist = (_n==1)
. list district re_rural se_rural re_urban se_urban if district <= 10 & tolist,
> sep(0)
```

	district	re_rural	se_rural	re_urban	se_urban
1.	1	-.9432691	.3156852	-.558359	.2332665
118.	2	-.0427011	.3822029	.0017684	.493834
138.	3	-.0149571	.6242161	.2263715	.4698407
140.	4	-.2864846	.3990107	.5869046	.3988538
170.	5	.0688648	.3102899	.0046016	.4685461
209.	6	-.3979315	.2762392	.2761181	.4204175
274.	7	-.1910399	.4046772	.0079117	.4938647
292.	8	.034071	.3180097	.2266263	.4689558
329.	9	-.3737992	.3923573	.0764026	.4569863
352.	10	-.5640147	.4769353	.0233582	.4939753

◀

### □ Technical note

When these data were first introduced in [ME] **melogit**, we noted that not all districts contained both urban and rural areas. This fact is somewhat demonstrated by the random effects that are nearly 0 in the above. A closer examination of the data would reveal that district 3 has no rural areas, and districts 2, 7, and 10 have no urban areas.

The estimated random effects are not exactly 0 in these cases because of the correlation between urban and rural effects. For instance, if a district has no urban areas, it can still yield a nonzero (albeit small) random-effects estimate for a nonexistent urban area because of the correlation with its rural counterpart.

Had we imposed an independent covariance structure in our model, the estimated random effects in the cases in question would be exactly 0. □

### □ Technical note

The estimated standard errors produced above with the `reses()` option are conditional on the values of the estimated model parameters:  $\beta$  and the components of  $\Sigma$ . Their interpretation is therefore not one of standard sample-to-sample variability but instead one that does not incorporate uncertainty in the estimated model parameters; see *Methods and formulas*.

That stated, conditional standard errors can still be used as a measure of relative precision, provided that you keep this caveat in mind. □

## ▷ Example 3: Obtaining predicted probabilities

Continuing with [example 2](#), we can obtain predicted probabilities, the default prediction:

```
. predict p
(predictions based on fixed effects and posterior means of random effects)
(option mu assumed)
(using 7 quadrature points)
```



These predictions are based on a linear predictor that includes both the fixed effects and the random effects due to district. Specifying the `conditional(fixedonly)` option gives predictions that set the random effects to their prior mean of 0. Below we compare both over the first 20 observations:

```
. predict p_fixed, conditional(fixedonly)
(option mu assumed)
. list c_use p p_fixed age child* in 1/20
```

	c_use	p	p_fixed	age	child1	child2	child3
1.	no	.3572114	.4927182	18.44	0	0	1
2.	no	.21293	.3210403	-5.56	0	0	0
3.	no	.4664207	.6044016	1.44	0	1	0
4.	no	.4198625	.5584863	8.44	0	0	1
5.	no	.2504834	.3687281	-13.56	0	0	0
6.	no	.2406963	.3565185	-11.56	0	0	0
7.	no	.3572114	.4927182	18.44	0	0	1
8.	no	.4984106	.6345998	-3.56	0	0	1
9.	no	.4564025	.594723	-5.56	1	0	0
10.	no	.465447	.6034657	1.44	0	0	1
11.	yes	.2406963	.3565185	-11.56	0	0	0
12.	no	.1999512	.3040173	-2.56	0	0	0
13.	no	.4498569	.5883406	-4.56	1	0	0
14.	no	.439278	.5779263	5.44	0	0	1
15.	no	.4786124	.6160359	-0.56	0	0	1
16.	yes	.4457945	.584356	4.44	0	0	1
17.	no	.21293	.3210403	-5.56	0	0	0
18.	yes	.4786124	.6160359	-0.56	0	0	1
19.	yes	.4629632	.6010735	-6.56	1	0	0
20.	no	.4993888	.6355067	-3.56	0	1	0

◀

#### ► Example 4: Intraclass correlations for higher-level models

Continuing with [example 2](#), we can also compute intraclass correlations for the model using `estat icc`; see [\[ME\] estat icc](#).

In the presence of random-effects covariates, the intraclass correlation is no longer constant and depends on the values of the random-effects covariates. In this case, `estat icc` reports conditional intraclass correlations assuming 0 values for all random-effects covariates. For example, in a two-level model, this conditional correlation represents the correlation of the latent responses for two measurements on the same subject, both of which have random-effects covariates equal to 0. Similarly to the interpretation of intercept variances in random-coefficient models ([Rabe-Hesketh and Skrondal 2012](#), chap. 16), interpretation of this conditional intraclass correlation relies on the usefulness of the 0 baseline values of random-effects covariates. For example, mean centering of the covariates is often used to make a 0 value a useful reference.

Estimation of the conditional intraclass correlation in the Bangladeshi contraceptive study setting of [example 2](#) is of interest. In [example 2](#), the random-effects covariates `rural` and `urban` for the random level `district` are mutually exclusive indicator variables and can never be simultaneously 0. Thus we could not use `estat icc` to estimate the conditional intraclass correlation for this model, because `estat icc` requires that the random intercept is included in all random-effects specifications.

Instead, we consider an alternative model for the Bangladeshi contraceptive study. In [example 2](#) of [\[ME\] melogit](#), we represented the probability of contraceptive use among Bangladeshi women with fixed-effects for urban residence (`urban`), age (`age`), and the number of children (`child1`–`child3`). The random effects for urban and rural residence are represented as a random slope for urban residence and a random intercept at the district level.

We fit the model with `melogit`:

```
. use http://www.stata-press.com/data/r15/bangladesh, clear
(Bangladesh Fertility Survey, 1989)
. melogit c_use urban age child* || district: urban, covariance(unstructured)
(iteration log omitted)
Mixed-effects logistic regression          Number of obs   =   1,934
Group variable:      district              Number of groups =    60
                                         Obs per group:
                                         min =           2
                                         avg =          32.2
                                         max =          118
Integration method: mvaghermite           Integration pts. =    7
                                         Wald chi2(5)    =   97.50
Log likelihood = -1199.315                 Prob > chi2     =   0.0000
```

c_use	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
urban	.8157875	.1715519	4.76	0.000	.4795519	1.152023
age	-.026415	.008023	-3.29	0.001	-.0421398	-.0106902
child1	1.13252	.1603285	7.06	0.000	.818282	1.446758
child2	1.357739	.1770522	7.67	0.000	1.010723	1.704755
child3	1.353827	.1828801	7.40	0.000	.9953882	1.712265
_cons	-1.71165	.1605618	-10.66	0.000	-2.026345	-1.396954
<b>district</b>						
var(urban)	.6663237	.3224689			.258074	1.720387
var(_cons)	.3897448	.1292463			.203473	.7465413
<b>district</b>						
cov(urban, _cons)	-.4058861	.1755414	-2.31	0.021	-.7499408	-.0618313

LR test vs. logistic model:  $\chi^2(3) = 58.42$       Prob >  $\chi^2 = 0.0000$   
 Note: LR test is conservative and provided only for reference.

We use `estat icc` to estimate the intraclass correlation conditional on urban being equal to 0:

```
. estat icc
Conditional intraclass correlation
```

Level	ICC	Std. Err.	[95% Conf. Interval]	
district	.1059201	.0314045	.058246	.1849518

Note: ICC is conditional on zero values of random-effects covariates.

This estimate suggests that the latent responses are not strongly correlated for rural residents (`urban == 0`) within the same district, conditional on the fixed-effects covariates.

► Example 5: Estimating the residual intraclass correlation

In example 4 of [ME] **melogit**, we fit a three-level model for the cognitive ability of schizophrenia patients as compared with their relatives and a control. Fixed-effects covariates include the difficulty of the test, **difficulty**, and an individual's category, **group** (control, family member of patient, or patient). Family units (**family**) represent the third nesting level, and individual subjects (**subject**) represent the second nesting level. Three measurements were taken on all but one subject, one for each difficulty measure.

We fit the model with **melogit**:

```
. use http://www.stata-press.com/data/r15/towerlondon
(Tower of London data)
. melogit dtlm difficulty i.group || family: || subject:
  (iteration log omitted)
Mixed-effects logistic regression           Number of obs   =           677
```

Group Variable	No. of Groups	Observations per Group		
		Minimum	Average	Maximum
family	118	2	5.7	27
subject	226	2	3.0	3

```
Integration method: mvaghermite           Integration pts. =           7
                                           Wald chi2(3)    =          74.90
Log likelihood = -305.12041                Prob > chi2     =           0.0000
```

dtlm	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
difficulty	-1.648505	.1932075	-8.53	0.000	-2.027185	-1.269826
group						
2	-.2486841	.3544076	-0.70	0.483	-.9433102	.445942
3	-1.052306	.3999921	-2.63	0.009	-1.836276	-.2683357
_cons	-1.485863	.2848455	-5.22	0.000	-2.04415	-.9275762
family						
var(_cons)	.5692105	.5215654			.0944757	3.429459
family>						
subject						
var(_cons)	1.137917	.6854853			.3494165	3.705762

```
LR test vs. logistic model: chi2(2) = 17.54           Prob > chi2 = 0.0002
Note: LR test is conservative and provided only for reference.
```

We can use **estat icc** to estimate the residual intraclass correlation (conditional on the difficulty level and the individual's category) between the latent responses of subjects within the same family or between the latent responses of the same subject and family:

```
. estat icc
Residual intraclass correlation
```

Level	ICC	Std. Err.	[95% Conf. Interval]	
family	.1139105	.0997727	.0181851	.4715289
subject family	.3416307	.0889471	.192923	.5297291

`estat icc` reports two intraclass correlations for this three-level nested model. The first is the level-3 intraclass correlation at the family level, the correlation between latent measurements of the cognitive ability in the same family. The second is the level-2 intraclass correlation at the subject-within-family level, the correlation between the latent measurements of cognitive ability in the same subject and family.

There is not a strong correlation between individual realizations of the latent response, even within the same subject.



## Methods and formulas

Methods and formulas for predicting random effects and other statistics are given in *Methods and formulas* of [ME] **meglm postestimation**.

## References

- De Backer, M., C. De Vroey, E. Lesaffre, I. Scheys, and P. De Keyser. 1998. Twelve weeks of continuous oral therapy for toenail onychomycosis caused by dermatophytes: A double-blind comparative trial of terbinafine 250 mg/day versus itraconazole 200 mg/day. *Journal of the American Academy of Dermatology* 38: S57–S63.
- Lesaffre, E., and B. Spiessens. 2001. On the effect of the number of quadrature points in a logistic random-effects model: An example. *Journal of the Royal Statistical Society, Series C* 50: 325–335.
- Rabe-Hesketh, S., and A. Skrondal. 2012. *Multilevel and Longitudinal Modeling Using Stata*. 3rd ed. College Station, TX: Stata Press.

## Also see

- [ME] **melogit** — Multilevel mixed-effects logistic regression
- [ME] **meglm postestimation** — Postestimation tools for meglm
- [U] **20 Estimation and postestimation commands**